

Logische Funkschoonen

Logische Funkschoonen köönt Fälle uteneenhollen. So köönt Se Bereken abhängig vun en Bedingen dörföhrn. Woans dat geht, lehrt Se in düät Kapittel.

Woso eentlich „logische Funkschoonen“? Eentlich weer doch allens, wat wi maakt hebbt, logisch. Ja, avers dat Woort „logisch“ meent nu en strenge Ja-Nee-Wenn-Denn-Logik. Logische Funkschoonen sünd Funkschoonen, bi de Ja orr Nee, Wahr orr Flsch rutkümmt. – Huch?

Verglieken

Fangt wi mit dat Eenfachste an. Geevt Se düsse Formel in. Wat maakt de Formel denn? Se feddert Excel op, den Inholl vun de twee Zelln op Gliekheit to prøven!

	A	B	C
1	1	2	=A1=B1
2			

So is en Vergliek opbuut:

- Dat Gliekheitsteken seggt Excel, dat dat hier wat maken schall.
- Denn de eerste Zell (orr de eerste Vergliekspartner),
- de Verglieksoperator un
- de tweete Zell (orr de tweete Vergliekspartner).

In Klaarext also etwa „Hallo Excel, kiek na, wat A1 gliek B1!“

	A	B	C
1	1	2	FALSCH
2			

Un dat Ergeevnis? Falsch natürlich. Sogoor **FALSCH!** In Grootbookstaven also. **FALSCH** is en **logische Weert**.

	A	B	C
1	2	2	WAHR
2			

Nu ännert Se den Inholl vun de Vergliekszelln so, dat se gliek sünd. Foorts stellt Excel jüst dat fast. Ok in Grootbookstaven. Ok **WAHR** is en logische Weert.

Un dat sünd denn ok al de twee eenzigen logischen Werte, de dat giff. (Excel nöönt ehr ok Wahrheitswerte.) Düsse logischen Werte kaamt in’n Alldag ok faken vör: jümmers da, ’neem een nich „villicht“ orr „en beten“ orr „deelwies“ seggen kann. Bi Exaams giff dat bloots bestahn orr dörfallen, nix dortwüsch; un as de Volksmund weet, kannst ok nich „en beten swanger“ ween, bloots Ja orr Nee, entweder – oder. Op düsse simple Logik sünd alle logischen Funkschoonen vun Excel opbuut.

Een seggt ok för...	
JA	NEE
WAHR	FALSCH
1	0
TRUE	FALSE

Dat giff avers noch mehr Verglieksoperatoren. De Tabell list ehr op. (De Teken > un < sitt links nerrn op de Tastatur!) Se köönt Excel fragen, wat Vergliekspartner grötter, lütter orr ungliek sünd. Un jümmers mutt Excel Ja orr Nee, WAHR orr FALSCH seggen.

Verglieksoperatoren	
=	gliek
>	grötter as
>=	grötter orr gliek
<	lütter as
<=	lütter orr gliek
<>	ungliek

De Verglieken to testen, leggt Se sik de nerrn stahn Tabell an. De Tallenreegn vun 10 bet 0 un vun 0 bet 10 maakt Excel för Se, wenn Se em de eersten Tallen vorgeevt (süh Siet 19). De Formeln köönt Se ja överall nerrn kopeern.

	A	B	C	D	E	F	G
1	Zahl 1	Zahl 2	gleich?	ungleich?	Zahl 1 größer?	Zahl 1 größer oder gleich?	Zahl1 * Zahl2 größer als Zahl1 + Zahl2?
2							
3	10	0	=A3=B3	=A3<>B3	=A3>B3	=A3>=B3	=(A3*B3)>(A3+B3)
4	9	1	=A4=B4	=A4<>B4	=A4>B4	=A4>=B4	=(A4*B4)>(A4+B4)
5	8	2	=A5=B5	=A5<>B5	=A5>B5	=A5>=B5	=(A5*B5)>(A5+B5)
6	7	3	=A6=B6	=A6<>B6	=A6>B6	=A6>=B6	=(A6*B6)>(A6+B6)
7	6	4	=A7=B7	=A7<>B7	=A7>B7	=A7>=B7	=(A7*B7)>(A7+B7)
8	5	5	=A8=B8	=A8<>B8	=A8>B8	=A8>=B8	=(A8*B8)>(A8+B8)
9	4	6	=A9=B9	=A9<>B9	=A9>B9	=A9>=B9	=(A9*B9)>(A9+B9)
10	3	7	=A10=B10	=A10<>B10	=A10>B10	=A10>=B10	=(A10*B10)>(A10+B10)
11	2	8	=A11=B11	=A11<>B11	=A11>B11	=A11>=B11	=(A11*B11)>(A11+B11)
12	1	9	=A12=B12	=A12<>B12	=A12>B12	=A12>=B12	=(A12*B12)>(A12+B12)
13	0	10	=A13=B13	=A13<>B13	=A13>B13	=A13>=B13	=(A13*B13)>(A13+B13)
14							

Un dat kümmt rut. Prööv't Se't in'n Kopp na. Stimmt, orr?

	A	B	C	D	E	F	G
1	Zahl 1	Zahl 2	gleich?	ungleich?	Zahl 1 größer?	Zahl 1 größer oder gleich?	Zahl1 * Zahl2 größer als Zahl1 + Zahl2?
2							
3	10	0	FALSCH	WAHR	WAHR	WAHR	FALSCH
4	9	1	FALSCH	WAHR	WAHR	WAHR	FALSCH
5	8	2	FALSCH	WAHR	WAHR	WAHR	WAHR
6	7	3	FALSCH	WAHR	WAHR	WAHR	WAHR
7	6	4	FALSCH	WAHR	WAHR	WAHR	WAHR
8	5	5	WAHR	FALSCH	FALSCH	WAHR	WAHR
9	4	6	FALSCH	WAHR	FALSCH	FALSCH	WAHR
10	3	7	FALSCH	WAHR	FALSCH	FALSCH	WAHR
11	2	8	FALSCH	WAHR	FALSCH	FALSCH	WAHR
12	1	9	FALSCH	WAHR	FALSCH	FALSCH	FALSCH
13	0	10	FALSCH	WAHR	FALSCH	FALSCH	FALSCH
14							

Frietied un Weder

Nu schall Excel mit sin logischen Funkschoonen wat „entscheiden“ – dat schall, je nadem, wa dta Weder is, uns Planen för de Frietied maken.

Dat Weder

- Leggt Se sik düsse lütte Tabell an. De bunten Zelln warrt wi glieks mit logische Werte orr logische Funkschoonen fülln!

	A	B	C	D	E
1	Wetter			Was tun wir?	
2	Sonnenschein			Fahrrad-Ausflug	
3	bewölkt			Kino	
4	Regen			Rodeln	
5	Schnee			Drachen steigen lassen	
6	windig				
7					

- Nu schrievt Se, je na de Wederlaag, in de Zell B2 in, wat de Sün schient. Schrievt Se einfach **wahr** orr **falsch** in de Zell.
- Sodraad Se **Enter** drückt, wannelt Excel de Ingaav in Grootbookstaven üm! Dat erkennt also den logischen Weert, so, as dat ok Funkschoonsnaams erkennt.
- Nu bruukt wi de eerste echte logische Funkschoon. Is dat bewölkt? Wi seggt einfach: Wenn de Sün schient, kann dat nich bewölkt ween – un annersrüm. De twee Angaven sluut enanner ut. In düsse Zell schall jümmers denn **automatisch** WAHR stahn, wenn bi Sünnschien FALSCH steiht, un annersrüm. Wi bruukt also en Funkschoon (ja!), de den logischen Weert ut Zell B2 in sin Gegendeel verkehrt!

	A	B
1	Wetter	
2	Sonnenschein	falsch

	A	B	C
1	Wetter		
2	Sonnenschein	FALSCH	Fahrrad
3	bewölkt	=nicht(B2)	Kino
4	Regen	NICHT(Wahrheitswert)	
5	Schnee		Drach

- De **Nicht**-Funkschoon is de rechte för düssen Zweck! Ehr eenzige Opgaav is dat, ut WAHR FALSCH to maken un annersrüm. Draagt Se also in de Zell B3 in =nicht(B2). Klickt Se dorbi in de Klammer einfach op de Zell B2.

- Excel schrifft allens vun sülvst groot; dat erkennt also den Funkschoonsnaam. (Dat Lüttschrieven is en gode Prööv, denn wenn Se en Tippfehler maakt harrn, wöör Excel en Funkschoonsnaam nich erkennen un dat ok nich ümwanneln!) Dat Ergeevnis is – dat Gegendeel vun den Weert in Zell B2!
- Nu ännert Se mal den Weert bi **Sonnenschein!** De **Bewölkt**-Zell schalt automaatsch üm!
- Füllt Se den Rest vun de Wederzells ut, ahn en anner Funkschoon to bruken.

	A	B	C
1	Wetter		
2	Sonnenschein	FALSCH	
3	bewölkt	WAHR	
4	Regen		

De Frietied

- Se sünd en Fohrradfan. Jümmers wenn de Sün schient, wüllt Se en Fohrrad-Utflog maken. Also Sünnschien = Fohrradutflog. Wat mööt Se in düsse Zell indragen? Klar: de einfachste vun alle Formeln, de Gliekheit.

	A	B	C	D	E
1	Wetter			Was tun wir?	
2	Sonnenschein	FALSCH		Fahrrad-Ausflug	=B2
3	bewölkt	WAHR		Kino	

- Nu bruukt wi de neegste logische Funkschoon. In't Kino wüllt Se gahn, wenn't regent orr wenn't wulkig is. Een vun beed, orr beed tosaam, drievt Se vör de Kinolienwand.
- Hier bruukt wi de **Oder**-Funkschoon. Füllt Se düsse Zell so ut: =oder(B3;B4)

	A	B	C	D	E	F	G	H	I
1	Wetter			Was tun wir?					
2	Sonnenschein	FALSCH		Fahrrad-Ausflug	FALSCH				
3	bewölkt	WAHR		Kino	=oder(B3;B4)				
4	Regen	FALSCH		Rodeln	ODER(Wahrheitswert1; [Wahrheitswert2]; [Wahrheitswert3]; ...)				
5	Schnee	FALSCH		Drachen steigen lassen					

De Oder-Funkschoon

Düsse Funkschoon prüöv't all ehr Argumente, wat een dorvun WAHR is, un schalt denn foorts op WAHR. Bloots wenn **all** Argumente FALSCH sünd, is ok **Oder** falsch. **Oder** seggt also „tominnst een mutt WAHR ween; man een langt mi.“

Un so is de Funkschoon anlegt:

=Oder(Zell 1;Zell 2;Zell 3;....)

Dat is anners as in de Alldagsspraak: Normaal seggt wi ja „Zell 1 **oder** Zell 2 **oder** Zell 3...“ Bi Excel steiht dat **Oder** ganz vörn, un de Zellen folgt in de Klammer, mit Semikola dortwischen. Da muttst di eerstmal an wennen.

Dat lohnt avers, denn so sünd ok de restlichen logischen Funkschoonen anlegt. Sülstverständlich mööt in de Zellen, de de Argumente sünd, logische Werte stahn, also WAHR orr FALSCH.

Mehr Frietied

- Rodeln doot wi bloots, wenn ideaales Rodelweder is – also Sünnchien un Snee toglied! Welk Funkschoon bruukt wi hier? En Funkschoon, de bloots denn WAHR is, wenn Sünnchien un Snee beed WAHR sünd. Deit dat de **Oder**-Funkschoon? Nee, al wenn Sünnchien WAHR weer, weer de **Oder**-Funkschoon tofreden un wöör Se foorst dat Rodeln empfehlen. Man so ganz sünnner Snee rodelt sik dat ok in'n Sünnchien slecht, nich?
- Geevt Se de **Und**-Funkschoon in. De Zellen för Sünnchien un Snee sünd de Argumente.

	A	B	C	D	E	F	G	H
	Wetter			Was tun wir?				
1								
2	Sonnenschein	FALSCH		Fahrrad-Ausflug	FALSCH			
3	bewölkt	WAHR		Kino	WAHR			
4	Regen	FALSCH		Rodeln	=und(B2;B5)			
5	Schnee	FALSCH		Drachen steigen lassen	UND(Wahrheitswert1; [Wahrheitswert2]; [Wahrheitswert3]; ...)			
6	windig	WAHR						

De Und-Funkschoon

Se is jüst so anlegt as de **Oder**-Funkschoon:

=Und(Zell 1;Zell 2;Zell 3;....)

De wichtige Ünnerscheed is: **Und** is nienich tofreden, wenn nich **all** sin Argumenten toglied WAHR sünd. Bloots denn warrt ok **Und** WAHR. En einzig FALSCH langt, dat **Und** FALSCH warrt. Wa weer dat bi **Oder**? En einzig WAHR langt, un **Oder** warrt WAHR. **Und** is also de Pedant, wiess **Oder** ümgänglich un gau tofreden is.

Nevenwirken: **Und** mutt jümmers all sin Argumente dörgahn, denn wenn se all WAHR sünd, bloots dat letzte is FALSCH, denso gifft dü't letzte den Utslag, un **Und** warrt ok FALSCH. **Oder** mutt sin Argumente bloots so wiet dörkieken, bet date en findt, dat WAHR is. Dat kann ok dat allereerste Argument ween. Foorts is allens regelt, un **Oder** is WAHR.

- Geevt Se nu noch de letzte Formel in. Ehrn prächtigen Lenkdraken laat Se bloots flegen, wenn de Sünn schient un noog Wind da is. Beed toglied! Klassisches Drakenweder! Also wedder en Fall för **Und**.

	A	B	C	D	E
1	Wetter			Was tun wir?	
2	Sonnenschein	FALSCH		Fahrrad-Ausflug	FALSCH
3	bewölkt	WAHR		Kino	WAHR
4	Regen	FALSCH		Rodeln	FALSCH
5	Schnee	FALSCH		Drachen steigen lassen	=und(B2;B6)
6	windig	WAHR			
7					

- Nu speelt Se beten mit de Weder-Zelln rümmer un kiekte Se denn to, wa de Tabell Se jümmers wat anners för Ehr Frietied vörslait! Nich vergeten: Se dröfft bloots de Werte för Sünnschien, Regen, Snee un Wind ännern, avers nich de **Bewölkt**-Zell, denn de verännert sik automaatsch; da steiht ja de **Nicht**-Funkschoon in.

Naams

Stöört Se sik an den snaakschen, „unmenschlichen“ Buu vun de logischen Funkschoonen? Da köönt wi leider nix an maken, avers Se köönt dat vun en anner Siet her menschlicher maken. Brukt Se dat, wat Se lehrte hebbt, un geevt Se de Weder-Zelln Naams! Dorbi mööt Se natüürlich de Zelln nehmen, in de de logischen Werte staht. Klickt Se je op de Zell un denn in't Namenfeld **A** un nöömst Se de Zelln vun baven na nerrn *Sonne, Wolken, Regen, Schnee, Wind*.

Sonne		A	f _x FALSCH
	A	B	
1	Wetter		
2	Sonnenschein	FALSCH	
3	bewölkt	WAHR	

Nu löscht Se all de logischen Funkschoonen op de rechte Siet – un makt Se ehr nieg! Dat is en gode Öven. Bi't Klicken op de Weder-Zelln kaamt nu in de Formeln de Zellnaams vörtüüg! Da seht de Funkschoonen al veel „menschlicher“ ut.

	A	B	C	D	E
1	Wetter			Was tun wir?	
2	Sonnenschein	FALSCH		Fahrrad-Ausflug	=Sonne
3	bewölkt	=NICHT(Sonne)		Kino	=ODER(Wolken;Regen)
4	Regen	FALSCH		Rodeln	=UND(Sonne;Schnee)
5	Schnee	FALSCH		Drachen steigen lassen	=UND(Sonne;Wind)
6	windig	WAHR			
7					

Wenn

Dat süht doch nich richtig elegant ut, wenn da alltied düsse dicken WAHRs un FALSCHs in't Arbeitsblatt staht, nich? Kann Excel dat nich ok schöner? Dat kann't. Dorto bruukt wi nu de **Wenn**-Funkschoon.

De **Wenn**-Funkschoon is en logische Funkschoon; se mutt also logische Werte as Argumente kriegen, anners streikt se. Se is so anlegt:

=Wenn(Bedingung;Dann;Sonst)

Dat eentliche Argument is dorbi de **Bedingung**. Se mutt WAHR orr FALSCH ween. Dat kann ok de Adress vun en Zell ween, in de WAHR orr FALSCH steiht.

Dann un **Sonst** sünd Anwiesen, wat Excel in de jewieligen Fäll to doon hett:

- **Bedingung** = WAHR: **Dann** warrt maakt. **Sonst** warrt ignoreert.
- **Bedingung** = FALSCH: **Sonst** warrt maakt. **Dann** warrt ignoreert.

De **Wenn**-Funkschoon „twiegt“ also dör de Bedingung in twee Äst: Düt doon orr dat, Ja orr Nee. Dat is meist liek as in't richtige Leven.

Frietied mit Wenn


- Rückt wi wieder na rechterhand un leggt wi us noch en Spalt mit bunten Achtergrund an. Dor schall de Vörslag för de Frietied mit en fründlichen Satz staats mit en kolen FALSCH orr WAHR stahn. Dat deit de **Wenn**-Funkschoon.

	A	B	C	D	E	F	G	H	I	J	K
1	Wetter			Was tun wir?							
2	Sonnenschein	FALSCH		Fahrrad-Ausflug	FALSCH		=wenn(Sonne;"Dies ist ein schöner Tag zum Radfahren!";"Kein Fahrradwetter!")				
3	bewölkt	WAHR		Kino	WAHR		WENN(Prüfung; [Dann_Wert]; [Sonst_Wert])				
4	Regen	FALSCH		Rodeln	FALSCH						
5	Schnee	FALSCH		Drachen steigen lassen	FALSCH						
6	windig	WAHR									
7											

- Fangt wi an mit de **Wenn**-Funkschoon. För de **Bedingung** klickt Se op de Sonnenschein-Zell. Denn geevt Se dat Semikolon in. Keen Leerteken achter dat Semikolon!
- Is de Bedingung WAHR, denn schall de Text kamen, de dat Fohrradfohren vörsleit. Texte kaamt in Göösfööt! (Denkt Se an dat benutzerdefinierte Tallenformat, da weer't jüst so!)
- Nich vergeten: De Göösfööt an't Enn vun den Text! Denn wedder en Semikolon, keen Leerteken dorna!
- Denn de Text (in Göösfööt), de vun dat Fohrradfohren afraden deit. Nich vergeten: Mit Göösfööt afsluten! Denn Klammer to. **Enter**.

Göösfööt!

De lichtste Fehler is dat hier, de Göösfööt to vergeten. Wenn Text wiest warn schall, mööt de Göösfööt unbedingt ween. Wenn se bi **Dann** orr **Sonst** fehlt, meckert Excel ganz fürchterlich. Süh Bild. Denn weet Se, wat los is.



Verschachtelte logische Funkschoonen

Nu maakt wi't jüst so bi't Kino. Bi welk Bedingen wüllt wi in't Kino gahn? Bi Wolken orr Regen. De richtige **Oder**-Funkschoon steiht al in Zell E3. Momang! Dat is de Bedingen, un de Bedingen is al en Funkschoon? Ja, dat geht! Natüürlich geht dat!

Logische Funkschoonen köönt sülvst de **Bedingung** vun **Wenn**-Funkschoonen warrn, denn se ergeevt WAHR orr FALSCH; un de Bedingung mutt WAHR orr FALSCH weer, dat is wichtig. Süss kann se ween, wat se will.

Fangt wi nu de **Wenn**-Funkschoon an. Bi de **Bedingung** klickt Se einfach op de Zell **E3**. Denn maakt Se den **Dann**- un **Sonst**-Deel trecht.

	D	E	F	G	H
1	Was tun wir?				
2	Fahrrad-Ausflug	WAHR		Dies ist ein schöner Tag zum Radfahren!	
3	Kino	FALSCH		=wenn(E3;"Waschechtes Kinowetter!";"Ab ins Grüne!")	
4	Rodeln	FALSCH		WENN(Prüfung; [Dann_Wert]; [Sonst_Wert])	
5	Drachen steigen lassen	WAHR			

Dör düd warrt de Zell E3 sotoseggen to'n Twüschenschritt. Man düssen Twüschenschritt kannst ok insporen! Dorto bruukt Se bloots de Kuraasch, de Funkschoonen to verschachteln. Doot Se dat.

Kiekt Se sik de **Oder**-Funkschoon in Zell **E3** noch mal nau an. De mutt nau so in de **Bedingung** vun de **Wenn**-Funkschoon – blots ahn dat Gliekheitsteken. Gaht wi nu in Zell G3 un geevt wi de **Wenn**-Funkschoon in. Dorbi mööt Se nu scharp op de Klammern passen. Excel helpt natüürlih mal wedder, so veel dat kann: dat farvt de ünnerscheidlichen Klammernpoore in ünnerscheidlich Klöören. Dat binnere Klammernpoor is gröön, dat butere swatt. In uns Bild nerrn mutt de letzte swatte Klammer noch tippt warrn – süss meckert Excel!

	D	E	F	G	H	I	J
1	Was tun wir?						
2	Fahrrad-Ausflug	FALSCH		Kein Fahrradwetter!			
3	Kino	WAHR		=wenn(oder(Wolken;Regen);"Ab ins Kino!";"Ab ins Grüne!")			
4	Rodeln	FALSCH		WENN(Prüfung; [Dann_Wert]; [Sonst_Wert])			
5	Drachen steigen lassen	FALSCH					

So, nu noch de twee annern **Wenn**-Funkschoonen schrievn. Dat öövt, un mit de Öven geht langsam de geistige Muskelkater weg, den düsse Funkschoonen to Anfang jümmers maakt.

Peilt Se as eerst noch mal de logische Funkschoon an, de dat Rodeln regelt; denn schrievt Se de as Bedingen in düsse **Wenn**-Funkschoon.

Ok hier in dat Bild mutt bloots de letzte Klammer noch sett warrn.

F	G	H	I	J	K
	Kein Fahrradwetter!				
	Ab ins Kino!				
	=wenn(UND(Sonne;Schnee);"Rodelwetter!";"Der Schlitten bleibt im Keller!")				
	WENN(Prüfung; [Dann_Wert]; [Sonst_Wert])				

F	G	H	I	J
	Kein Fahrradwetter!			
	Ab ins Kino!			
	Der Schlitten bleibt im Keller!			
	=wenn(UND(Sonne;Wind);"Heute fliegt er wieder!";"Kein Drachenwetter!")			
	WENN(Prüfung; [Dann_Wert]; [Sonst_Wert])			

Liekso löppt dat bi't Drakenstiegen; ok hier is de Bedingen en **Und**-Funkschoon. In dat Bild linkerhand sünd alle Klammern korrekt sett.

So, un nu geneten!

- Lehnt Se sik trüch un kiekt Se sik nochmaal de verschachtelten logischen Funkschoonen an, de Se da schreven hebbt.
- Laavt Se sik, dat Se so vigeliensche Saken al trechtkregen hebbt!
- Maakt Se de Spalten D, E un F unsichtbor, dat bloots de Weder-Spalten un de Spalten mit de **Wenn**-Funkschoonen na sünd. (Spalten utblenden? Wa güng dat noch? Siet 54 in den Grundweten-Deel!) Se köönt de Spalten D, E un F sogoor löschen! Se hebbt ja de „Twüschenschreed“ inspoort.
- Speelt Se mit dat Weder rüm un geneet Se, wa de Funkschoonen arbeiten doot!

Vergliek as Bedingen

Gaht Se nu noch mal na de Arbeitsmapp orr Tabell, 'neem Se de Kosten vun dat grote Mittsommernacht-Fest berekent hebbt. Dor steiht ja ok de Gesamtbedrag vun de Kosten.

F	G
Kann ich mir dieses Fest leisten?	
Sparbuch	12.400,00 €
Einnahmen aus Verlosung (geschätzt)	4.000,00 €
Wettgewinn	18.000,00 €
	34.400,00 €

Föögt Se jichenswo in de Tabell düsse lütte To-satz-Reken in. Dorbi investeert Se Ehr ganzes Spoorgoothebben un en Wettgewinn – einfach allens, wat Se hebbt – in dat Fest. Dorünner tellt Se düsse Bedrääg tosamen.

Nu prüöv't Se mit enVergliek, wat düsse Summe grötter is (>) as de gesamte Kostenbedrag. Wiel dat Verglieken WAHR orr FALSCH ergeevt, köönt se ok Bedingen in en **Wenn**-Funkschoon ween! Test Se dat mit en Funkschoon, as se in dat nafolgen Bild bi **A** wiest is.

F	G	H	I
Kann ich mir dieses Fest leisten?			
Sparbuch	12.400,00 €		
Einnahmen aus Verlosung (geschätzt)	4.000,00 €		
Wettgewinn	18.000,00 €		
	34.400,00 €		
=WENN(G8>Gesamtbetrag;"Das Fest ist drin!";"Es fehlt noch was!")			
WENN(Prüfung; [Dann_Wert]; [Sonst_Wert])			

Wat passeert, wenn Se den Vergleichsoperator ännert – vun grötter (>) na lütter (<)? Geevt Se sik de Anter toerst in'n Kopp, denn porbeert Se't ut! – Richtig: Denn kümmt de **Wenn**-Funkschoon jüst annersrüm rut, un Se mööt **Dann** un **Sonst** vertuschen, dat dat wedder stimmt.

Bedrag utgeven

In den **Dann**- orr **Sonst**-Deel köönt nich bloots nette Texte stahn; da kannst ok reken

laten. Se köönt da t.B. ok den Differenzbetrag twüschen Gesamtkosten un Egenkapital wiesen laten. Dafür geevt Se einfach de Formel in – avers ahn Giekheitsteken!

=WENN(Gesamtbetrag<G8;"Das Fest ist drin!";Gesamtbetrag-G8)

För Nieschierige: Texte un Bedrääg

Mit en &-Teken köönt Se in den **Dann**- un **Sonst**-Deel sogoor Texte **un** Bedrääg utgeven laten! Dat & verknüppt Text (in Göösfööt) un Formeln. Versöckt Se mal de Formel in dat Bild nerrn. Verännert Se denn de Tallen in de Tabell, dat Se de Formel arbeiten sehn köönt!

=WENN(Gesamtbetrag<G8;"Es bleiben "&G8-Gesamtbetrag&" Euro übrig!";"Es fehlen noch "&Gesamtbetrag-G8&" Euro")

Reken mit Wenn

Nemt wi us nu mal en würrklich knifflige Opgaav vör:

- In en Firma kriggt jedeen Verköper, de en Opdrag an Land treckt, en Provischoon vun 3% vun den Opdragsweert.
- Wokeen avers en Opdrag an Land treckt, de mehr as 10.000 €inbringt, de kriggt as Bontje en Extra-Prämie vun 5% vun den Opdragsweert.

Düsse Regeln giff de nerrn stahn Tabell wedder.

	A	B	C	D	E	F
1	Provisionsabrechnung					
2						
3						
4	Auftragswert	12.000,00 €		Provisionssatz		3%
5	Provision	360,00 €		Prämie		5%
6	Extra-Prämie	600,00 €		Mindestauftragswert für Prämie		10.000,00 €
7						
8	Der Verkäufer erhält	960,00 €				
9						
10						

- Leggt Se de Tabell an.
- Sett Se in dat geel ünnerleggte Feld (Zell B6) de passen **Wenn**-Funkschoon in. Bloots wenn de Opdragsweert över de Grenz is, schall hier de Prämie stahn!
- Berekent Se de Provischoon (Zell B5) un amenn de Summe, de de Verköper altohoop kriggt – also Zell B8 in dat Bild baven.

Test Se dat Ganze, indem dat Se ünnerscheidlich Bedrääg as Opdragsweert (Zell B4 in dat Bild baven) henschrievt, orr indem dat Se den Provischoonssatz (Zell F4), de Prämie (Zell F5) orr den Mindestopdragsweert (Zell F6) verännert – orr all tosamen! Versöcht Se dat toerst alleen, ehrrdat Se de Lösungstipps bruukt!

Lösung:

- De Bedingen is en Vergliek twischen den Mindestopdragsweert, vun den af an dat de Extra-Prämie gifft, un den würllichen Opdragsweert.
- Wenn de würlliche Opdragsweert höger as de Mindestopdragsweert in Zell F6, denn kriggt de Mitarbeiter de Prämie (prozentual to'n Opdragsweert!), süß kriggt he 0 €
- So süht dat Ganze ut, wenn Se mit normale Zelladressen arbeiten doot:

	A	B	C	D	E	F
1	Provisionsabre					
2						
3						
4	Auftragswert	12000		Provisionsatz		0,03
5	Provision	=B4*F4		Prämie		0,05
6	Extra-Prämie	=WENN(B4>F6;B4*F5;0)		Mindestauftragswert für		10000
7				Prämie		
8	Der Verkäufer erhält	=SUMME(B5:B6)				
9						

- Schöner un verständlicher warrt dat natüürllich, wenn Se de Zellen Naams geven hebbt:

	A	B	C	D	E	F
1	Provisionsabrechnung					
2						
3						
4	Auftragswert	12000		Provisionsatz		0,03
5	Provision	=Auftragswert*Provision		Prämie		0,05
6	Extra-Prämie	=WENN(Auftragswert>Mindestauftragswert;Auftragswert*Prämie;0)		Mindestauftragswert für		10000
7				Prämie		
8	Der Verkäufer erhält	=SUMME(B5:B6)				
9						

Weer dat to simpel? Denn versöök Se nu düt:

De Verköper in de Firma kriegt nich bloots Provischoon un Prämien, sünnern natüürllich ok en fastes Gehalt. Nu schallt bloots de „armen“ Verköper, de ehr Fastgehalt ringer as 1500 €, de Extra-Prämien kriegen.

Buut Se düsse Bedingen in de Tabell in! Dat Bild verraadt de Lösung; avers eerst probeern!

	A	B	C	D	E	F	G	H
1	Provisionsabrechnung							
2								
3								
4	Auftragswert	12.000,00 €		Provisionsatz		3%		
5	Provision	360,00 €		Prämie		5%		
6	Extra-Prämie	- €		Mindestauftragswert für		10.000,00 €		
7				Prämie				
8	Der Verkäufer erhält	360,00 €		Gehaltsgrenze		1.500,00 €		
9				tatsächliches Gehalt		2.000,00 €		
10								

In de Bedingen mutt nu en **Und**-Funkschoon stahn, wieldat twee Bedingen toglied erfüllt ween mööt: dicker Opdrag **un** ringes Fastgehalt. De Verköper in dat Bild baven kriggt keen Prämie, ofschoonst he en dicken Opdrag an Land trocken hett – denn sin Fastgehalt (Zell F9) is to hooch. De kumplette Formel, de nu in Zell B6 stahn mutt, steiht in de Bearbeitungsliest! De Zell mit den Naam **Gehaltsgrenze** is Zell F8 in dat Bild, de Zell F9 hett den Naam **Gehalt**.

Se seht, mit logische Funkschoonen kannst orrig wat utrichten! Nahst maakt wi ehr noch op anner Rebetten nütt.